

The Mjølner System

BETA Compiler

Reference Manual

Mjølner Informatics Report

MIA 90-02(1.6)

October 1997

Copyright © 1990-97 Mjølner Informatics ApS.
All rights reserved.

No part of this document may be copied or distributed
without the prior written permission of Mjølner Informatics

Table of Contents

| | |
|---|-----------|
| 1 INTRODUCTION | 1 |
| 2 SIMPLE USE OF THE COMPILER | 2 |
| 3 THE BETA LIBRARY | 3 |
| 4 FILES GENERATED BY THE COMPILER | 5 |
| 5 IMPLEMENTATION DEVIATIONS | 6 |
| 5.1 The BETA Book | 6 |
| 5.2 Restrictions | 6 |
| 5.3 Extensions | 11 |
| 5.3.1 String Literals as References | 11 |
| 5.3.2 Special Characters in String Literals | 12 |
| 5.3.2 Simple If | 12 |
| 5.3.3 Xor Primitive | 12 |
| 5.4.4 Short-circuit Boolean Expressions | 12 |
| 5.4.5 ## for Objects | 13 |
| 6 THE FRAGMENT SYSTEM | 14 |
| 6.1 Fragment Language Syntax | 14 |
| 6.2 Fragment Denotations | 18 |
| 6.3 Fragment Properties | 18 |
| 6.4 Modularization of Data Structures | 20 |
| 6.4 Modularization with INNER | 21 |
| 6.5 Formal Syntax of Fragment Language | 23 |
| 6.6 File Name Restrictions | 24 |
| 7 ERROR HANDLING | 25 |
| 7.1 Syntax Errors | 25 |
| 7.2 Static Semantic Errors | 25 |

| | |
|---|-----------|
| 7.3 Check for bound SLOTS. | 26 |
| 7.4 Assembler and Linker Errors | 26 |
| 7.5 System Errors | 27 |
| 7.6 Run-time Errors | 27 |
| 8 COMPILER ARGUMENTS | 29 |
| 9 MACHINE DEPENDENT CONFIGURATIONS | 32 |
| 9.1 BUILD Property | 33 |
| Syntax: | 33 |
| Functionality | 34 |
| Example | 34 |
| More than one build pr. fragment | 34 |
| 10 CODE GENERATION FOR MULTIPLE MACHINES | 36 |
| 10.1 Placement of Object Code | 36 |
| 10.2 Macro Expansion | 36 |
| BIBLIOGRAPHY | 39 |
| APPENDIX A. SEMANTIC ERRORS AND WARNINGS | 41 |
| A.1 Semantic Errors | 41 |
| A.2 Semantic Warnings | 44 |
| APPENDIX B. THE BETA GRAMMAR | 45 |
| APPENDIX C. NEW FEATURES IN V5.3 | 49 |
| C.1 New Platforms | 49 |
| C2. The BUILD Property | 49 |
| INDEX | 51 |

1 Introduction

This manual describes version 5.3 of the BETA compiler (corresponding to release 4.1 of the Mjølner System). The compiler implements most parts of the BETA language as described in [MMN93]. There are, however, some implementation restrictions.

The user should read section 5 for a description of the implementation restrictions and deviations from [MMN93]

See Appendix C for an overview of new features in V5.3 of the compiler. The BETA compiler is accompanied by a large collection of libraries and application frameworks. This includes a text concept, and libraries for input/output on keyboard, screen and files, a user interface package, a library of well-known datastructures, and a meta-programming system. The Mjølner System is available for The Power Macintosh, UNIX workstations such as Sun-4 (SPARC running Solaris), HP-9000 series 700, Silicon Graphics running IRIX, and PC's running Windows NT, Windows 95 or Linux.

On Macintosh the user interface system is implemented on top of the Macintosh Toolbox. For Macintosh there is also a library that interfaces directly to the Toolbox.

On UNIX, the user interface system is implemented on top of the X Window System (X11R3 or later). A number of UNIX facilities can be accessed via a UNIX library.

On Windows 95 and Windows NT, the user interface system is implemented on top of WIN32.

A general interface to C and assembly language is part of the libraries/compiler.

The rest of this manual is organized as follows: Section 2 describes the simplest way of using the compiler. Section 3 describes the organization of the basic BETA libraries. Section 4 describes the files generated by the compiler. Section 5 describes various deviations in the implementation of BETA. Section 6 describes the implementation of the fragment system. Section 7 describes compile- and run-time errors. These sections contain useful information for all users.

The remaining sections are only for advanced users. In section 8, a number of different arguments to the compiler are described. In section 9, it is described how to instantiate machine dependent configurations of a program. In section 10 it is described how code is generated for multiple machines.

2 Simple Use of the Compiler

The following is an example of a very small BETA program.

A BETA program

```
ORIGIN '~beta/basiclib/v1.6/betaenv'
--- PROGRAM: descriptor ---
( #
do 'Welcome to Mjolner' -> putLine
#)
```

Only the part between (# ... #) is BETA. The ORIGIN specification:

ORIGIN

```
ORIGIN '~beta/basiclib/v1.6/betaenv'
```

describes that version 1.6¹ of the fragment `betaenv` from the BETA basic library (`basiclib`) is used.

The fragment name and category:

Program slot

```
--- PROGRAM: descriptor ---
```

describes that the BETA program is filled into a slot in `betaenv` called `PROGRAM`. The BETA compiler is integrated with the Mjølner fragment system. The above BETA program is an example of a BETA fragment.

Assume that the above BETA fragment is located in the file `foo.bet`. The BETA fragment may then be compiled by issuing the command

Compiling

```
beta foo.bet
```

which will compile and link³ the BETA fragment. The final object code will be in the file `foo`, which may be executed.

More information

How to invoke the compiler depends on whether Macintosh, PC or UNIX is used. Details about the different variants of the BETA compiler may be found in [MIA90-6] for Macintosh, [MIA94-32] for Windows, and in [MIA90-4] for UNIX.

¹ The actual version to be used depends on the current release installed at the available hardware.

² On most platforms, binary code is generated directly. In this case, the assembly phase is omitted.

³ On most platforms, binary code is generated directly. On the HP platform, the compiler generates assembly code which is assembled before linking.

3 The BETA Library

The BETA library is a collection of patterns and objects that include input/output, a text concept, the user interface toolkit, the metaprogramming system, a container library, a system library, etc. The library is organized as fragments.

One part of the library contains the basic patterns and objects which are used by most programs. This basic BETA library is called `basiclib` and is described in [MIA90-8], which also describes the interface to C and assembly language.

basiclib

The library `basiclib` contains a number of different fragments groups containing basic patterns, a text concept, various functions and control patterns, a file concept, etc. One of these fragment groups is `betaenv`, which contains the basic patterns, the text concept, other basic patterns and objects representing the screen and the keyboard. All BETA programs must use `betaenv`, which has the form:

```
(# ...
  (* A lot of useful patterns *)
  ...
  <<SLOT LIB: attributes>>
  ...
  program: <<SLOT program: descriptor>>
  theProgram: ^|program;

do ...
  &|program[] -> theProgram[];
  theProgram;
  ...
#)
```

betaenv

The `LIB` slot describes where most libraries are inserted. The `program` slot describes where an ordinary user program is inserted (see section 6 for more explanation of this).

On UNIX, the BETA library is often located in the directory `/usr/local/lib/beta`.

Location of libraries

For Macintosh, the convention is that the BETA library is located in a folder called `beta`.

In the rest of this manual, we assume that the basic library is located in `/usr/local/lib/beta`. We also use the UNIX convention for denoting directories with the character `/` to separate directory and file names.

The Mjølnær System contains directories for the various libraries. The basic library `basiclib` is e.g. located in the directory:

```
/usr/local/lib/beta/basiclib
```

The directory for a library contains directories corresponding to different versions of the library. Version 1.6 of `betaenv` is contained in the directory

```
/usr/local/lib/beta/basiclib/v1.6
```

This directory contains the fragment groups constituting `basiclib`. Instead of referring to a specific version, it is possible to refer to the current official version by means of the name `current`. (This is not possible on Windows and Macintosh).

The Mjølner System accepts the following abbreviation for the BETA library:

~beta

```
~beta      denotes      /usr/local/lib/beta
```

The meaning of `~beta` can be changed by using the `BETALIB` environment variable, see [MIA 90-04].

To sum up, the file containing the current version of `betaenv` may be referred to by:

```
~beta/basiclib/current/betaenv
```

A user-program using `betaenv` may then look as follows:

```
ORIGIN '~beta/basiclib/current/betaenv'
--- PROGRAM: descriptor ---
( #
do 'Welcome to Mjolner' -> PutLine
#)
```

Please note, that on Windows and Macintosh the separator in `ORIGIN` specifications is also `/.` See section 6.2.

Assume that the above program resides on the file `foo.bet`. The program may then be compiled by issuing the command:

```
beta foo.bet
```

The file `foo` will now contain an executable version of `foo.bet`.

When developing the program, it may be an advantage to invoke the compiler as

```
beta -r foo.bet
```

Repeating mode

This will run the compiler in repeating mode. After having translated the fragments specified in the argument list, if in repeating mode, the compiler prompts the user for the name of another fragment to be translated. Hitting `<RETURN>` in this case will recompile the program last compiled. See section 8 for a survey of the legal command line options.⁴

More information

Please consult the BETA tutorial [MIA 94-24] for a quick survey of the BETA language and the basic libraries.

⁴ This is currently not possible on Windows and Macintosh.

4 Files Generated by the Compiler

For each fragment file, a number of other files may be produced by the compiler; let `foo.bet` be a BETA fragment. Then

- `foo.lst` contains information about possible syntactic and static semantic errors. If such errors occur, then the file contains a pretty-print of the fragment with an indication of the error(s). See section 7 for further information about error handling. Possible semantic error messages are listed in appendix A.
- `foo.ast` or `foo.astL` contains the abstract syntax tree representation of the compiled source code for big-endian and little-endian architectures, respectively. The AST files are used by many tools in the Mjølner System.
- `foo.s` contains the generated assembly code for the compiled source code⁵. Assembly files are located in subdirectories named according to the machine type, to which the source code has been compiled. Currently, the directories `sun4s`, `hpux9pa`, `nti`, `linux`, `sgi`, and `ppc_mac` can be created. These directories are automatically created by the compiler, if not present already. The assembly file is usually deleted by the compiler after assembly.
- `foo.o` contains the object code generated by the assembler. Like `foo.s`, this file is placed in a subdirectory.
- `foo.db` contains information used by the debugger Valhalla when debugging the `foo` fragment. See [MIA 92-12]. Like `foo.s`, this file is placed in a subdirectory.

List files

Abstract syntax tree files

Assembler files

Object files

Debug files

The above list of files is generated for each fragment group that is included in a program. In addition, the following two files are generated for each program:

- `foo` containing the executable code for the program.
- `foo.job` containing directives for assembly and linking. Like `foo.s`, this file is placed in a subdirectory. This file is usually deleted by the compiler after linking.

Executable

Job file

For some implementations (e.g. Windows NT) other extensions than `.s` and `.o` may be used.

⁵ On most platforms, binary machine code is generated directly. In this case, no assembly file is generated.

5 Implementation Deviations

5.1 The BETA Book

The BETA book

The BETA language is described thoroughly in [MMN93].

It is prerequisite to be familiar with [MMN93] in order to use the Mjølnir System.

This book is currently the only definition of the BETA language, but a precise language definition is being worked on.

A short introduction to BETA and the Mjølnir System may also be found in the Mjølnir Tutorial [MIA 94-24] and in [KLMM94].

The BETA grammar is given in appendix B.

There are a few of deviations from [MMN93] in the current implementation of BETA. These deviations are described below.

5.2 Restrictions

Restrictions

1. The `integer` operations `+`, `-`, `*`, `div`, `mod`, `=`, `<>`, etc. will work on 32 bits.
2. Assignment between instances of `integer`, and `real` is allowed. In assignments of reals to integers the values are truncated.

Assignment between instances of `integer` and `char` is allowed. Character constants have their ASCII char value. Assignment of an arbitrary `integer` value to `char` instances may thus give meaningless results.

Assignment between instances of `integer` and `boolean` is allowed, but will give a warning. In a future release these assignments will not be allowed and will give an error. The patterns `true` and `false` have the values 1 and 0 respectively. Assignment of an arbitrary `integer` value to `boolean` instances may thus give meaningless results.

The following table shows legal combinations of operands and the result type.

Binary operators: <, <=, >, >=

| | int | char | real | bool | sref |
|------|------|------|------|------|------|
| int | bool | bool | bool | * | * |
| char | bool | bool | bool | * | * |
| real | bool | bool | bool | * | * |
| bool | * | * | * | bool | * |
| sref | * | * | * | * | bool |

4. Arithmetic/logical operators.

The following tables show legal combinations of operands and the result type. The notation is explained in item 2 above.

Binary operators: +, -, *, div

| | int | char | real |
|------|------|------|------|
| int | int | int | real |
| char | int | int | * |
| real | real | * | real |

Binary operator: mod

| | int | char |
|------|-----|------|
| int | int | int |
| char | int | int |

Binary operator: /

| | int | char | real |
|------|------|------|------|
| int | real | real | real |
| char | real | real | * |
| real | real | * | real |

char is likely to be eliminated as a legal operand for / in a future version.

Binary operators: and, or, xor

| | bool |
|------|------|
| bool | bool |

Unary operators: + -

int, char, real result type is the same as operand type

Unary operator: not**bool** result type is **bool**5. In `if`-imperatives

```
(if E0 // E1 then ... // E2 then ... if)
```

the exit-lists of `E0`, `E1`, `E2`, ... must consist of exactly one integer, real, char, boolean or reference.

6. Inserted items, i.e.,

```
do ...; P; ...
```

([MMN93], section 5.10.2) are implemented as dynamic items (`&P`). However, the user is urged to use dynamic items for recursion in order to ensure compatability with future releases.

7. Inserted components, i.e.,

```
do ...; |(# ... #); ...
```

([MMN93], section 5.10.3) have not been implemented.

8. Virtual superpatterns, i.e.,

```
A::< (# ... #); (* Where A is some virtual *)
B: A(# ... #)
```

have not been implemented.

By using a *final binding*, this problem may often be overcome like this:

```
A:: (# ... #); (* A is no longer virtual *)
B: A(# ... #)
```

The situation may also occur in a more indirect way:

```
graph:
  (# node:< (# ... #);
    nodeList: @list(# element::< node #);
    ...
  #);
```

Here the virtual further binding of `element` in `list` is not allowed, since `node` is itself virtual.

The current version of the compiler will allow final binding using a pattern that is itself virtual. That is, you can do this:

```
graph:
  (# node:< (# ... #);
    nodeList: @list(# element:: node #);
    ...
  #);
```

General virtual prefixes behave much like multiple inheritance and will not be implemented in the near future.

9. The labelled compound imperative

```
A: (L: imp1; imp2; ...; impN :L)
```

has been eliminated from the language. Instead the following construct may be used:

```
A: (# do imp1; imp2; ... ; impN; #)
```

Inserted items with no declarations and no superpattern will be inlined in the enclosing code. There will thus be no execution overhead compared to the old (never implemented) labelled compound imperative statement.

10. Consider the following example:

```
A: (# X: ^P; (* reference to item qualified by P *)
    B: ^|P (* reference to component qualified by P *)
    do ...
      this(P)[] -> X[]; (* legal use of this(P)[] *)
      this(P)[] -> R[]; (* illegal use of this(P)[] *)
    #)
```

The illegal use is due to the fact that `this(P)[]` is considered a reference to an item object and not a component object.

11. In declarations like:

```
P: <AD>(# ... #);
X: @<AD>;
Y: ^<AD>;
```

it is checked that `<AD>` is a *static* denotation, where *static* is defined as follows:

- A name `A` is always static
- In a remote-name `R.A`, `R` must be a static object
- Use of `THIS(A).T` is static
- Only in `Y: ^P.T`, can `P` be a pattern
- Denotations using `R[e]`, and `(foo).bar` are *not* static

This means that e.g. descriptors like:

```
R[e].A(# ... #)
(foo).bar(# ... #)
R.P(# ... #) where 'R' is a dynamic ref.
```

are only allowed in *imperatives*.

For `Y: ^R.P` where `R` is a dynamic reference, the compiler will currently report a warning and suggest to use

```
Y: ^A.P where A is the qualification of R.
```

Note: that when using `--noWarnQua`, this warning will *not* be printed. A future release may change the warning to an error.

12. There are some deviations with respect to the implementation of concurrency. Please consult [MIA90-8] before using the concurrency.
13. It is in general not possible to use `leave P` or `restart P` where `P` is a pattern. `P` must in general be a label. *However*, the following has been implemented:

```
P: (#
  do
  ...
  leave P;
  ...
  restart P;
  ...
#)
```

Leave/restart from an inserted item, however, is *not* supported by the current version of the compiler:

```
P: (#
  do
  ...
  (#
  do
  ...
  leave P; (* ILLEGAL *)
  ...
  restart P; (* ILLEGAL *)
  ...
#)
...
#)
```

14. A pattern where the object descriptor is described as a slot cannot be used as a super-pattern. I.e. the following is illegal:

```
A: <<SLOT Pdesc: descriptor>>;
B: P(# ... #); (* illegal *)
```

Instead the following can often be used:

```
C: (# do <<SLOT Pdesc: descriptor>> #)
D: P(# ... #); (* legal *)
```

15. The `Program` pattern as described in the chapter on exception handling in [MMN93] has not been implemented.
16. There are some restrictions on the use of fragments as described in section 6 below.

5.3 Extensions

5.3.1 String Literals as References

The pattern `Text` enters and exits a char-repetition. This means, that a text may be initialized using constant strings as follows:

```
t: @text;
do 'hello' -> t;
```

Many operations involving texts, however, takes *references* to texts as enter/exit parameters. This is mainly for efficiency reasons.

To allow easy invocation of such operations on string literals, the following is *also* allowed:

Extensions

```

t: ^text;
do 'hello' -> t[];

```

The semantics of this is, that a text object is instantiated, initialized by the constant string, and finally assigned to the text reference.

5.3.2 Special Characters in String Literals

The following special characters are allowed in BETA string literals.

| | | | |
|-----------------|------------------------|-------------------|---------------|
| <code>\a</code> | alert (bell) character | <code>\v</code> | vertical tab |
| <code>\b</code> | backspace | <code>\\</code> | backslash |
| <code>\f</code> | formfeed | <code>\?</code> | question mark |
| <code>\n</code> | newline | <code>\'</code> | single quote |
| <code>\r</code> | carriage return | <code>\"</code> | double quote |
| <code>\t</code> | horizontal tab | <code>\ooo</code> | octal number |

Notice that you may now use `\'` as an alternative to `'` to include a literal quote in a string. E.g.: `'Tom\'s Cottage'`. This has the consequence, though, that to type the backslash character, you must do it as: `'\\'`.

`\ooo` can also be `\o` or `\oo`, provided that the character immediately following it is not a digit.

5.3.2 Simple If

Often the following If statement is used:

```

b: @boolean;
do (if b//TRUE
  then ...
  else ...
  if);

```

The current version of the compiler supports an extension to the BETA language called Simple If. This extension means, that the case-selector `//` may be omitted, if the evaluation on the left hand side exits a boolean. That is, the above may be written

```

b: @boolean;
do (if b
  then ...
  else ...
  if);

```

Like in the general `if`-statement, the `else` part is optional.

5.3.3 Xor Primitive

An `xor` primitive is supported as a basic operation on booleans. That is

```

b1, b2, b3: @boolean
do b1 xor b2 -> b3

```

is possible.

5.4.4 Short-circuit Boolean Expressions

Boolean expressions are implemented as short-circuit.

That is, in

B1 or B2 B2 is *not* evaluated if B1 is true

B1 and B2 B2 is *not* evaluated if B1 is false

5.4.5 ## for Objects

You may use P## as an alternative to P._struc, when P is an *object*. Previously ## was only allowed for *patterns*.

6 The Fragment System

The Mjølner System is based on the notion of fragment. The fragment system must be used for splitting a large program into smaller units (fragments). The fragment system is used to support modularization, separation of interface and implementation parts, variant control and separate compilation. It is highly recommended to use the fragment system, since this may improve the structure of the program.

The principles of the fragment system are described in [MMN93]. In the following it is assumed that the reader is familiar with this description.

The description in [MMN93] is slightly more idealized than the actual implementation in the Mjølner System:

In [MMN93], the syntax of the fragment language is given in terms of diagrams. The fragment language implemented by the Mjølner System has a textual syntax.

Implemented
categories

In the Mjølner System, slots have only been implemented for the syntactic categories <<DoPart>>, <<object-Descriptor>> and <<attributes>>.

Attributes
restrictions

A fragment form of the category <<attributes>>, may only contain pattern declarations. It cannot contain any other kind of declarations, including virtual pattern declarations, virtual pattern bindings, static or dynamic declarations.

The alias `descriptor` can be used instead of `objectDescriptor`.

In the rest of this section, details of the Mjølner System implementation of fragments are given.

Fragment group

In the current system, fragments are organized in groups. A group is stored as a file. The BETA compiler accepts a BETA program in the form of one or more files. Each file must contain a group of fragments (i.e. one or more fragments).

6.1 Fragment Language Syntax

In the following some of the examples of fragments from [MMN93] will be given followed by the syntax used by the Mjølner System. The first example shows the simplest possible BETA fragment-group:

| |
|--|
| NAME 'mini1' |
| ORIGIN 'betaenv' |
| PROGRAM: descriptor |
| (# do 'Hello world!' -> PutLine #) |

Graphical syntax

The fragment-group is stored in the file `mini1.bet`, which is also the name of the fragment-group. The following syntax is used by the Mjølnér System:

```
ORIGIN '~beta/basiclib/v1.6/betaenv'
-- program: descriptor --
(#
do 'Hello world!' ->PutLine
#)
```

Textual syntax

The origin `betaenv` has been expanded into a complete file name for `betaenv`.

The next example is an example defining a library fragment:

| |
|---|
| NAME 'mylib' |
| ORIGIN 'betaenv' |
| LIB: attributes |
| Hello: (# do 'Hello' -> PutText #); World: (# do 'World' -> PutText #) |

Library

This fragment is stored in a file `mylib.bet` and the corresponding syntax in the Mjølnér System is:

```
ORIGIN '~beta/basiclib/v1.6/betaenv'
-- LIB: attributes --
Hello: (# do 'Hello' -> PutText #);
World: (# do 'World' -> PutText #)
```

The following fragments is an example of a fragment including the above defined library:

| |
|--------------------------------------|
| NAME 'mini2' |
| ORIGIN 'betaenv' |
| INCLUDE 'mylib' |
| PROGRAM: descriptor |
| (# do Hello; World; newLine #) |

Using the library

This fragment is stored in a file `mini2.bet` and has the following syntax:

```
ORIGIN '~beta/basiclib/v1.6/betaenv';
INCLUDE 'mylib';
-- program: descriptor --
(#
do Hello; World; newLine
#)
```

The following example shows a fragment with a body:

Body

| |
|--|
| NAME 'textlib' |
| ORIGIN 'betaenv' |
| INCLUDE 'mylib' |
| LIB: attributes |
| SpreadText: {A blank is inserted between all chars in the text 'T'} (# T: @text enter T <<SLOT SpreadText:DoPart>> exit T #); |
| BreakIntoLines: {'T' refers to a text which is to be split into lines.} {'w' is the width of the lines.} (# T: ^ Text; w: @ Integer enter(T[,w) <<SLOT BreakIntoLines: DoPart>> #) |

It is stored in a file `textlib.bet` and has the following syntax:

```

ORIGIN '~beta/basiclib/v1.6/betaenv';
BODY 'textlibbody';
---LIB: attributes---
SpreadText:
(* A blank is inserted between all chars in the text 'T' *)
(# T: @text
enter T
<<SLOT SpreadText: DoPart>>
exit T
#);
BreakIntoLines:
(* 'T' refers to the text to be split into lines. *)
(* 'w' is the width of the lines. *)
(# T: ^ Text; w: @ Integer
enter(T[,w)
<<SLOT BreakIntoLines: DoPart>>
#)

```

The body of `textlib` is shown in the next example:

| |
|---|
| NAME 'textlibbody' |
| ORIGIN 'textlib' |
| SpreadText: DoPart |
| do (# L: @integer do (for i: (T.length->L)-1 repeat (' ',L-i+1) -> T.InsertCh for) #) |
| BreakIntoLines: DoPart |
| do T.scan (# sepInx,i,l: @integer; do i+1->i; l+1->l; (if (ch<=' ') then i->sepInx if); (if l=w then (nl,sepInx)->T.InxPut; i-sepInx->l if); #); T.newline; |

This fragment is stored in a file `textlibbody.bet`. The corresponding syntax is:

```

ORIGIN 'textlib'
-- Spreadtext: DoPart --
do (# L: @Integer
    do ...
    #)
--BreakIntoLines: DoPart --
do ...

```

Notice, that when local variables are needed in a DoPart slot, it may be necessary to make an inserted item in the DoPart. Alternatively a Private descriptor slot may be declared in the interface, and the L attribute moved to the Private fragment, which should then be placed in textlibbody.bet too.

Finally a general outline of a fragment group with several include, body and fragments is shown in the next example:

**General
fragment file
structure**

| |
|------------|
| NAME F |
| ORIGIN G |
| INCLUDE A1 |
| INCLUDE A2 |
| ... |
| INCLUDE Am |
| BODY B1 |
| BODY B2 |
| ... |
| BODY Bk |
| F1: S1 |
| ff1 |
| F2: S2 |
| ff2 |
| ... |
| Fn: Sn |
| ffn |

This fragment group is stored in a file F.bet and the syntax is:

```

ORIGIN 'G';
INCLUDE 'A1' 'A2' ... 'Am';
BODY 'B1' 'B2' ... 'Bk';
Prop1; Prop2; ... Prop1
-- F1: S1 --
ff1
-- F2: S2 --
ff2
...
-- Fn: Sn --
ffn

```

Prop1, Prop2, ..., Prop1 are *properties* that may be defined for a fragment. Formally the ORIGIN, INCLUDE, and BODY parts are also properties. In section 6.3 a list of possible properties is given.

6.2 Fragment Denotations

In the examples above, terms like

```
INCLUDE '~beta/basiclib/v1.6/betaenv'
```

Fragment
denotation

were used. Below we will use the term `FragmentDenotation` for the “fragment path” given in, e.g., the `INCLUDE` property. The other properties, that accept `FragmentDenotations` as arguments are explained in section 6.3.

Notice that a `FragmentDenotation` is *not* the same as a file name, although it resembles a UNIX file path, and although it normally corresponds directly to a (set of) file(s):

'/' separator

1. The separator in the `FragmentDenotation` is always the '/' character, e.g., also for BETA programs on the Macintosh, where ':' is used for *file* paths.
2. As explained in section 3, the notation '~beta' is legal in `FragmentDenotations` on all platforms, and simply means “the place BETA is installed”. As mentioned, the meaning of '~beta' can be controlled by using the `BETALIB` environment variable, please consult [MIA 90-04], [MIA 94-34], and [MIA 90-06] for details.
3. The notation '.' means 'current directory/folder' on all platforms, and the notation '..' means 'father directory/folder', i.e. the directory containing a given directory.
4. It is not allowed to specify an extension (e.g. '.bet' or '.ast') in a `FragmentDenotation`.

~beta

There are some restrictions in the legal fragment *file* names, which also apply to the `FragmentDenotations`, please see section 6.6.

6.3 Fragment Properties

The fragment system allows arbitrary properties to be associated with fragments. The BETA compiler recognizes the following properties: For most users, only `ORIGIN`, `INCLUDE`, and `BODY` are relevant.

ORIGIN *<TextConst>*

The origin of a fragment is a fragment which is used when binding fragment-forms to slots.

INCLUDE *<StringList>*

Specifies one or more fragments that are always included when using this fragment.

BODY *<StringList>*

Specifies one or more fragments that fills the slots in this fragment file, but are not visible.

MDBODY *<MachineSpecificationList>*

Specifies one or more machine dependent fragments that fills the slots in this fragment file dependent on the machine type. See section 9 for further description.

BUILD *<MachineSpecificationList>*

The BUILD property is used to specify rules for keeping external (i.e. non-BETA) sources up to date, and to include the external files in the link directive. The BUILD property unifies the OBJFILE and MAKE properties. See also section 9.

OBJFILE *<MachineSpecificationList>*

The object file is included in the linker-directive. This is typically an External library which is interfaced to via the External interface described in [MIA90-8]. See also BUILD and section 9.

BETARUN *<MachineSpecificationList>*

The standard BETA run-time system is replaced with the one in the object-file. See also section 9.

MAKE *<MachineSpecificationList>*

Specifies one or more makefiles to be executed before linking. See also section 9. The Makefile is executed relative to the directory, where the file containing the MAKE property is placed. See also BUILD

RESOURCE *<MachineSpecificationList>*

Specifies one or more resource files to be included in the application. Only used on Macintosh and Windows NT platforms. See also section 9.

LIBFILE *<MachineSpecificationList>*

Is similar to OBJFILE, but specifies inclusion of a library. See also section 9.

LINKOPT *<MachineSpecificationList>*

Machine dependent options to append to link directive for programs using the fragment. Only used on UNIX platforms. See also section 9.

ON *n1 n2 ... nk*

The compiler switches *n1 n2 ... nk* (positive numbers) are set. See also section 8.

OFF *n1 n2 ... nk*

The compiler switches *n1 n2 ... nk* (positive numbers) are cleared. See also section 8.

The terms *<MachineSpecificationList>*, *<StringList>*, and *<TextConst>* are syntactically explained in section 6.5.

6.4 Modularization of Data Structures

This section gives some advices that can be used when modularizing data structures. Consider the following program library (`stack.bet`):

```

ORIGIN '~beta/basiclib/v1.6/betaenv'
--- Lib: attributes ---
stack:
  (# element:< object;
   A: [100] ^element;
   top: @integer;
   push:
     (# e: ^element;
      enter e[]
      do top+1->top;
      e[] -> A[top][];
      #);
   pop:
     (# e: ^element;
      do A[top][] -> e[];
      top-1->top;
      exit e[]
      #);
   top:
     (# e: ^element;
      do A[top][]->e[];
      exit e[]
      #);
  #)

```

If we want to separate the interface and the implementation, this can be modularized in the following way:

Separating the interface

Introduce the following SLOTS:

```

ORIGIN '~beta/basiclib/v1.6/betaenv';
BODY 'stackImpl'
--- Lib: attributes ---
stack:
  (# element:< object;
   private: @<<SLOT private: descriptor>>;
   push:
     (# e: ^element;
      enter e[]
      <<SLOT pushBody: DoPart>>
      #);
   pop:
     (# e: ^element;
      <<SLOT popBody: DoPart>>
      exit e[]
      #);
   top:
     (# e: ^element;
      <<SLOT topBody: DoPart>>
      exit e[]
      #);
  #)

```


Create a new fragment file `stackImpl.bet`:

```
ORIGIN 'stack';
-- private: descriptor --
(# A: [100] ^element;
  top: @integer;
#)
-- pushBody: DoPart --
do private.top+1->private.top;
  e[] -> private.A[private.top][];
-- popBody: DoPart --
do private.A[private.top][] -> e[];
  private.top-1->private.top;
-- topBody: DoPart --
do private.A[private.top][]->e[]
```

The reason why the data representation (`A` and `Top`) is put into a descriptor slot instead of an attributes slot is that attributes slots may only contain patterns, no static items (objects) or object references. This is due to the implementation of separate compilation. Therefore it is necessary to put static items into an attribute (in this case `private`) that is declared by means of a descriptor slot. Because of this all accesses to the representation must be done via the `private` variable (see `pushBody`, `popBody` and `topBody`). Notice that the parameters are visible in the interface. If the operations had local variables they should not be shown in the interface.

6.4 Modularization with INNER

Programs fragments with do-parts that contain an `INNER` imperative e.g.:

```
ORIGIN '~beta/basiclib/v1.6/betaenv';
--- lib: attributes ---
A: (# do impl; imp2; INNER; imp3 #)
```

can be modularized in the following two ways depending on whether the `INNER` imperative should be visible in the interface or not.

If the `INNER` is preferred visible in the interface, the interface fragment could look like (`fooLib1.bet`):

```
ORIGIN '~beta/basiclib/v1.6/betaenv';
BODY 'fooImpl1'
-- lib: attributes --
A: (#
  do <<SLOT imp12slot: descriptor>>;
    INNER;
    <<SLOT imp3slot: descriptor>>
#)
```

Interface

and the implementation fragment (`fooImpl1.bet`):

```
ORIGIN 'fooLib1'
-- imp12slot: descriptor --
(# do impl; imp2 #)
-- imp3slot: descriptor --
(# do imp3 #)
```

Implementation

In this case a `DoPart` slot might be used instead (`fooLib2.bet`):

Using DoPart slot

```

ORIGIN '~beta/basiclib/v1.6/betaenv';
BODY 'fooImpl2'
-- lib: attributes --
A: (# <<SLOT impl2slot: DoPart>> #)

```

with the implementation fragment (`fooImpl2.bet`):

```

ORIGIN 'fooLib2'
-- impl2slot: DoPart --
do impl; imp2; INNER; imp3

```

Using do-parts like this, then although the `INNER` is not visible in the interface, the `A` pattern may still be specialized and behave as if the `INNER` was in the interface. Notice, that when specializing a pattern with no `INNER` in the do-part, the compiler will normally complain about this. But when the pattern being specialized contains a `SLOT`, the compiler will assume, that the `SLOT` contains an `INNER`. Thus it is possible to specialize the `A` pattern in `fooLib2`.

But if the `INNER` imperative is placed “inside” some structure e.g.:

```

A: (#
  do (if E1
      // E2 then INNER
      // E3 then imp
      if)
  #)

```

you might not want to show the `if` imperative in the interface. In this case a variant of the `INNER` construct may be used, in which case the interface fragment could be (`fooLib3.bet`):

```

ORIGIN '~beta/basiclib/v1.6/betaenv';
BODY 'fooImpl3'
--- lib: attributes ---
A: (# do <<SLOT Abody: descriptor>> #);

```

and the implementation fragment (`fooImpl3.bet`):

```

ORIGIN 'fooLib3'
--- Abody: descriptor ---
(#
do (if E1
    // E2 then INNER A
    // E3 then imp
    if)
#)

```

If a “normal” `INNER` had been used instead of `INNER A`, it would mean that specializations of the pattern containing the `INNER` in the do-part combine the actions at this point. But the pattern containing the `INNER` in the do-part, in this case would be the anonymous pattern in the `ABody` descriptor fragment. By using `INNER A`, it is ensured, that the control flow descends to the specialization of `A` although the `INNER` is inside the `ABody` descriptor.

A `DoPart` slot could also be used here, as in the previous example.

6.5 Formal Syntax of Fragment Language

The formal syntax of the BETA fragment-system is:

```

<TranslationUnit> ::= <Properties> <FormPart>
<FormPart> ::= * <FormDef>
<FormDef> ::= -- <FormDefinition>
<FormDefinition> ::= | <DescriptorForm> | <AttributesForm> | <dopart_form>
<DescriptorForm> ::= <NameDcl> : descriptor -- <ObjectDescriptor>
<AttributesForm> ::= <NameDcl> : attributes -- <Attributes>
<DopartForm> ::= <NameDcl> : dopart -- <DoPart>

<Properties> ::= <PropertyList>
<PropertyList> ::= + <PropertyOpt> ';'
<PropertyOpt> ::= ? <Property>
<Property> ::= |
    | <ORIGIN>
    | <INCLUDE>
    | <BODY>
    | <MDBODY>
    | <BUILD>
    | <OBJFILE>
    | <LIBFILE>
    | <LINKOPT>
    | <BETARUN>
    | <MAKE>
    | <RESOURCE>
    | <ON>
    | <OFF>
    | <Other>
<ORIGIN> ::= 'ORIGIN' <TextConst>
<INCLUDE> ::= 'INCLUDE' <StringList>
<BODY> ::= 'BODY' <StringList>
<MDBODY> ::= 'MDBODY' <MachineSpecificationList>
<BUILD> ::= 'BUILD' <MachineSpecificationList>
<OBJFILE> ::= 'OBJFILE' <MachineSpecificationList>
<LIBFILE> ::= 'LIBFILE' <MachineSpecificationList>
<LINKOPT> ::= 'LINKOPT' <MachineSpecificationList>
<BETARUN> ::= 'BETARUN' <MachineSpecificationList>
<MAKE> ::= 'MAKE' <MachineSpecificationList>
<RESOURCE> ::= 'RESOURCE' <MachineSpecificationList>
<ON> ::= 'ON' <IntegerList>
<OFF> ::= 'OFF' <IntegerList>
<StringList> ::= + <TextConst>
<IntegerList> ::= + <IntegerConst>
<MachineSpecificationList> ::= + <MachineSpecification>
<MachineSpecification> ::= <Machine> <StringList>
<Machine> ::= | <NameApl> | <Default>
<Default> ::= 'default'
<Other> ::= <NameDcl> <PropertyValueList>
<PropertyValueList> ::= * <PropertyValue>
<PropertyValue> ::= <Value>
<Value> ::= | <NameDcl> | <IntegerConst> | <TextConst>
<NameDcl> ::= <NameDecl>
<NameApl> ::= <NameAppl>
<TextConst> ::= <String>
<IntegerConst> ::= <Const>

```

**Fragment
Grammar**

**Property
Grammar**

Note that the symbol `--` may consist of two or more dashes (`-`), and that the old style `INCLUDE` and fragment syntax (`--INCLUDE fragment`) are not described by this grammar. This old-style `INCLUDE` syntax is no longer supported..

6.6 File Name Restrictions

Because of implementations details, the current version of the fragment system imposes the following restrictions on file names used for BETA programs.

1. It is not allowed for a program to use two files with the same name, say `foo.bet` (ignoring case), which both contains fragments of category `Attributes`.
2. It is not allowed for a program to use a file named, say, `foo.bet`, if `foo.bet` contains a fragment of category `Attributes`, and if there is a `SLOT` of category `ObjectDescriptor/Descriptor` or `DoPart` named `foo` in any of the files involved in the program. Again case is irrelevant.
3. It is not allowed to use the `'-'` (dash) character in fragment file names.
4. Because the `FragmentDenotation` separator character is `'/'` it is not allowed to use the `'/'` in fragment file names, not even on platforms where the file system would allow it.
5. In general, it is advisable to restrict the characters used in the fragment file names to be: `a-z`, `A-Z`, `0-9`, and `'_'`. If other characters are used in the fragment file names, there is a danger, that the supporting tools (such as linkers) will complain.

'-' is illegal in file names

Symptoms

The symptom on breaking rule 1 or 2 is typically a “Multiple defined symbol `M1FOO`” and the like, in the linking phase, the symptom for breaking rule 3 is that the compiler / Valhalla [MIA 92-12] / Sif [MIA 90-11] may become confused. Finally the symptom on breaking rule 5 may be a complaint from the assembler about illegal characters.

Except for rule 3, these restrictions only apply to the *file* names. The *directories / Folders* containing the files, may be freely named.

7 Error Handling

BETA programs containing errors will cause error messages during compilation. Error messages may appear during syntax analysis, static semantic analysis, code generation and assembly/linking. In addition various forms of system errors may occur.

7.1 Syntax Errors

A syntax error is given when there are errors in the context free syntax of the BETA program. These includes missing semicolons, non-matching brackets, etc. Such errors are printed on the screen and may look as follows:

```
Parse errors
# 1 ORIGIN '~beta/basiclib/v1.6/betaenv'
# 2 --PROGRAM: descriptor--
# 3 (# T: (# #);
# 4   X: [100) @integer;
# ***** ^
# Expected symbols: >= mod < <= = % <> > -> * ] div + /
xor or and
File "syntaxerror.bet"; Line 4
# 3 (# T: (# #);
# 4   X: [100) @integer;
# 5 do (for i: X.range repeat
# 6     3->X[i];
# 7   if)
# ***** ^
# Expected symbols: _NAME_ _KONST_ _STRING_ none not @@
restart leave ; (# % & (this + inner for tos suspend
File "syntaxerror.bet"; Line 7
```

The error message shows that there are syntax errors in lines 4 and 7. In line 4 the arrow(^) points at the place where an illegal symbol is met. The compiler gives a list of acceptable symbols. In this case) should have been a]. In line 7, the `if` should have been a `for`.

7.2 Static Semantic Errors

Static semantic errors appear in situations where a name is used without being declared, where a pattern name is used as an object, etc. Each error found is printed on the screen with a small indication of the context. After the checking, a pretty print of

the fragment including a precise indication of the error is generated on the `lst`-file (see section 4)⁶

In appendix A, the semantic error messages that may be reported by the compiler are listed.

7.3 Check for bound SLOTS.

In general the compiler will only attempt to link, if a PROGRAM slot has been found in the dependency graph.

If SLOTS of category `DoPart` or `Descriptor` in the dependency graph are not bound, and linking would otherwise have happened, the compiler issues a warning, and does not attempt to link.

Likewise, if two or more fragments tries to bind the same SLOT, the compiler will give a warning.

7.4 Assembler and Linker Errors

Errors may also appear during assembling and linking. The following type of errors may appear:

- The assembler/linker complains about a corrupt `.s` or `.o` file. This may happen if the compilation/assembly has been interrupted for some reason leaving an incomplete file. This can usually be handled by forcing a recompilation of the corresponding BETA file. (Delete the `.s` and `.o` files in question)
- The linker may report errors such as "Undefined Reference" or "Multiply Defined Symbol". This may be due to violations of the restrictions mentioned in section 6.
- The disk may run full during assembling or linking. Restart compilation after having obtained more disk space.

See also section 6.6.

⁶ Some semantic errors may cause the compiler to fail without generating a pretty print. There should however always be an error indication on the screen. In case the compiler fails during checking and it is not obvious for what reason, it is possible to trace the checking of declarations and imperatives using the option `--traceCheck` (see section 8). However, this may generate a large amount of output on the screen. The compiler may also fail during code generation. These errors may be traced using option `--traceCode`. However, tracing errors in this way should rarely be needed.

7.5 System Errors

Two kinds of system errors may appear: (1) Errors in the compiler, and (2) error situations in the operating systems. Most times a meaningful error message is given in these situations, but due to the nature of these errors this is not always the case.

Compiler errors should be reported to Mjølnér Informatics ApS. This can be done in one of three ways:

1. **Via electronic mail using the Internet address
support@mjolner.dk**
2. **By sending a fax to Mjølnér Informatics ApS at
+45 86 20 12 22**
3. **By issuing an ordinary mail to the address
Mjølnér Informatics
Science Park Aarhus,
Gustav Wieds Vej 10
DK-8000 Århus C
Denmark**

Operating system errors are often due to local problems. Examples of such errors may be: insufficient access to files, no more disc space, file server inaccessible, etc.

7.6 Run-time Errors

Run-time errors are errors in the program detected during its execution. In this case an error message is given and a dump of the call stack of objects is generated on the file `foo.dump` if the program is named `foo`.

Consider the following fragments (note that the name of the fragments are complete UNIX file paths)

/usr/smith/mylib.bet:

```
ORIGIN '~beta/basiclib/v1.6/betaenv'  
--LIB: attributes--  
lib1: (# do INNER #);  
lib2: lib1  
  (# T: (# x: @integer #);  
   R: ^T  
  do (* &T[]->R[] *)  
    l1l->R.x; (* R[] is NONE *)  
  INNER  
  #);  
lib3: lib2(# do 'hello'->putLine #)
```

/usr/smith/runtimeerr.bet:

```

ORIGIN '~beta/basiclib/v1.6/betaenv';
INCLUDE 'mylib'
--PROGRAM: descriptor--
(# foo1: (# do foo2 #);
  foo2: (# do foo.foo3 #);
  foo: @(# foo3: (# do lib3 #)#)
do foo1
#)

```

Execution of this program on a sun4 machine will result in the following runtimeerr.dump file:

.dump file

```

Beta execution aborted: Reference is none.

Call chain: (sun4)

  item lib3#<lib2#>lib1# in /usr/smith/mylib
    -- BETAENV~ in ~beta/basiclib/v1.6/betaenv
  item <foo3#> in /usr/smith/runtimeerr
    -- foo# in /usr/smith/runtimeerr
  item <foo2#> in /usr/smith/runtimeerr
    -- PROGRAM~ in /usr/smith/runtimeerr
  item <foo1#> in /usr/smith/runtimeerr
    -- PROGRAM~ in /usr/smith/runtimeerr
  comp <PROGRAM~> in /usr/smith/runtimeerr

basic component in ~beta/basiclib/v1.6/betaenv

```

The information in runtimeerr.dump has the following meaning:

- The activation stack of invoked objects is shown. Each element of the stack is shown as two lines. The object and its statically enclosing object.
- For each object, the name of the file where it is defined is also shown.
- From the above file it can be seen that the error occurred in an instance of lib3. The description lib3#<lib2#>lib1 shows the superpattern chain of lib3. The braces (<,>) indicates that the error occurred in the do-part of lib2.
- The symbol immediately after the name of an object shows its kind. The different possibilities are:

```

#   The descriptor belongs to a pattern, e.g. P: (#...#)
~   Singular named descriptor, e.g. X: @(# ... #)
*   Singular unnamed descriptor, e.g. ...; (# ... #);...
-   Descriptor SLOT.

```

Notice that, e.g. the PROGRAM SLOT is marked with both - and ~ since a descriptor SLOT gives rise to a singular named descriptor.

- It can be seen that lib3 was called from foo3, which was called from foo2, which was called from foo1, etc. The bottom most objects are defined in betaenv
- For each active object its enclosing object is shown, on a line starting with "--" The encloser of e.g. foo3 is foo. The rest of the objects have enclosers, which are slots.
- For each object, the corresponding fragment file is shown. The pattern lib3 is defined in the file
/usr/smith/mylib

8 Compiler Arguments

When activating the BETA compiler, the following command line arguments are valid.

Most options have both a "--<name>" and a "--no<name>" form: Activate the option using "--<name>"; deactivate the option using "--no<name>". In the listing below, the activating form is shown first (and explained), if both exist for an option.

For most options, there is a short (one-character) option for the non-default form. One-character options allow multiple option characters after the "-" (e.g. "-qwd").

Long option names are case insensitive, whereas one-character options are case sensitive.

Shortcuts

Case sensitiveness

A star (*) in the listings below indicates the *default* option.

--help **-h** Show a brief overview of the legal command line options

--repeat **-r** Run compiler in repeating mode. After having translated the fragments specified in the argument list, if in repeating mode, the compiler prompts the user for the name of another fragment to be translated:

```
Type Fragment File Name:
```

This interaction is continued until the compiler is explicitly killed, e.g. by sending a `control-C` or the end-of-stream character to the compiler process.

The compiler may also be given additional options at the prompt, e.g. you may type `--nolink foo.bet` to translate `foo.bet`, but avoid linking of it.

If no new fragments are specified at the prompt, the compiler will retranslate the last fragment it has translated when `<RETURN>` it typed.

By using repeating mode, the compiler saves time when analyzing dependencies between fragments, since fragments are saved in memory between compilations.

--noRepeat *

--link * Link program

--noLink **-x**

--static Use static linking

--dynamic * Use dynamic linking

--list * Generate .lst file, if semantic errors

--noList **-l**

| | | |
|-----------------------|-----------|---|
| --debug | * | Generate debug info to enable debugging. This is used by the BETA debugger—Valhalla. On the other hand, using <code>--noDebug</code> forces the linker to strip the application, which reduces the size of the executable files by 30-50%, and also speeds up linking time. The actual machinelevel code generated for the BETA program is identical with or without debug info. |
| --noDebug | -d | |
| --code | * | Generate code |
| --noCode | -c | |
| --checkQua | * | Generate runtime checks for QUA errors |
| --noCheckQua | -Q | |
| --checkNone | * | Generate runtime checks for NONE references |
| --noCheckNone | -N | |
| --checkIndex | * | Generate runtime checks for repetition index out of range |
| --noCheckIndex | -I | |
| --warn | * | Generate warnings |
| --noWarn | -w | |
| --warnQua | * | Generate warnings about runtime QUA checks |
| --noWarnQua | -q | |
| --verbose | | Verbose compiler info output |
| --quiet | * | Only compiler info on parse, check, etc. |
| --mute | | No compiler info output |
| --traceCheck | | Trace the compiler during semantic checking |
| --noTraceCheck | | |
| --traceCode | | Trace the compiler during code generation |
| --noTraceCode | | |
| --out | -o | Specify name to use for resulting executable image |
| --preserve | -p | Preserve generated .job and assembly files |
| --noPreserve | * | |
| --job | * | Execute the .job file |
| --noJob | -j | |
| --switch | -s | Set/unset one or more compiler switches. The <code>-s</code> option makes it possible to define one or more so-called compiler switches. Switches are specified as integers on the command line after <code>--switch</code> or <code>-s</code> , possibly terminated by a 0 (zero). Switches are used for a number of purposes: parameterization of the compiler, debugging, testing etc. The most interesting switches with respect to |

parameterization are listed below; notice that some of them may also be set as ordinary options.

- **5:** Suppress code generation. I.e. only semantic checking is performed. This switch will also set switch 33. Same as **-c**.
- **6:** Suppress linking. Same as **-x**.
- **14:** Do not generate run-time checks for NONE-references. Same as **-N**
- **15:** Do not generate run-time checks for index-errors. Same as **-I**.
- **18:** Preserve assembly- and job-files. Same as **-p**.
- **19:** Suppress notification of insertion of run-time checks for qualification errors in reference assignment. Same as **-q**.
- **21:** Continue translation after semantic errors.
- **23:** Preserve job-files.
- **32:** Do not produce .lst file in case of semantic errors. Same as **-l**.
- **33:** Do not execute .job file. Same as **-j**.
- **37:** Do not generate debugging information. Same as **-d**.
- **42:** Do not generate run-time checks for qualification errors in reference assignment. Same as **-Q**.
- **191:** Print each descriptor just before it is checked.
- **192:** Print each declaration just before it is checked.
- **193:** Print each imperative just before it is checked.
- **308:** Print each declaration just before code is generated for it.
- **311:** Print each imperative just before code is generated for it.

--linkOpts

Specify text string to be appended to the link directive

fragment1 ... fragmentN

Arguments other than the above mentioned options are treated as the names of fragments to be translated by the compiler. It should be noted that for an option to take effect in the translation of a fragment whose name is passed as argument to the compiler, the option must appear *before* the fragment name in the argument list.

9 Machine Dependent Configurations

In this section, the terminology of the fragment system is used freely without further explanation. The fragment system has been extended to support generic software descriptions. The same generic software description may be used to instantiate configurations for different machines. The term “machine” covers a CPU and an operating system running on that CPU.

Generic properties

The concept of generic software descriptions is implemented by means of special “generic properties”. Normally, a property has exactly *one* associated set of values. A generic property has *a number of* such value-sets. The idea is that the programmer can specify a value-set for each machine. These value-sets are the ones termed *<MachineSpecificationList>* in the formal specification of properties in section 6.3 and 6.5. As an example:

```
OBJFILE  sun4s    'xlib.o'
         linux    'zlib.o'
         default  'wlib.o'
```

Configurations

OBJFILE is the name of a generic property. The OBJFILE property is used for inclusion in the linkage phase of external object files, e.g. produced by a C compiler. A generic property specification should be seen as a kind of “switch/case” statement. The semantics of the above OBJFILE property is that when instantiating a configuration for the machine `sun4s`, the value `xlib.o` is chosen. This means that the object file `xlib.o` is included when linking a configuration for a `sun4s` machine. Similarly for `linux` machines. The `default` literal indicates that when instantiating configurations for machines *other* than `sun4s` or `linux`, the object file `wlib.o` should be included.

Besides OBJFILE, there are the following generic properties: MAKE, BETARUN, LIBFILE, LINKOPT, RESOURCE, and MDBODY. For all of these properties, the relation between machine symbols and value-sets are specified in the same manner as described above. To be precise, the following algorithm is used when instantiating a configuration for a specific machine type, say A.

1. If A matches any of the machine symbols of the generic property, the value-set associated with that particular machine symbol is chosen. If no match is possible, proceed with step 2.
2. If the symbol `default` is specified as machine symbol, the associated value-set is chosen. If not, a warning is issued.

The only distinction between the different generic properties is in the interpretation of the elements of the chosen value-set. For OBJFILE, the value-set is interpreted as external object files. MAKE is meant to point out a number of so-called makefiles. These are executed just prior to the linkage phase. A makefile is often used to keep the included object files up to date with respect to the source files from which they originate. For BETARUN, the value-sets must contain exactly one element, and this

element denotes the runtime system to be used in the resulting configuration. With respect to LIBFILE, the elements of the value-sets are interpreted as external libraries, e.g. the X11 library, to be included in the linkage phase. The chosen value-set in an MDBODY property denotes ordinary BETA fragments to be treated as if they had been specified by means of a normal BODY property. The MDBODY property may thus be used to specify that a fragment appears in a number of machine dependent variants. Finally, the LINKOPT property denotes arguments to append to the link-directive in the linking phase of compilations. Finally, the RESOURCE property is used (only on PC and Macintosh) to specify a set of resource files to add to the application.

Configurations are instantiated by the compiler, by default for the machine on which the compilation takes place. It is possible to instantiate a configuration for a machine other than the one, on which the compilation is performed (“cross-compilation”). This requires extensions to the Mjølner System; please contact Mjølner Informatics if this is needed.

Cross-compilation

9.1 BUILD Property

The BUILD property unifies the OBJFILE and MAKE properties. The BUILD property is used to specify rules for keeping external (i.e. non-BETA) sources up to date, and to include the external files in the link directive.

Syntax:

```
BUILD <machine> '<objectfile>'
                '<dep1>' '<dep2>' ... '<depN>'
                '<command>';
```

where

| | |
|--------------|--|
| <machine> | is the target machine specification (see MDBODY description) |
| <objectfile> | is the external objectfile to include and possibly maintain. A \$\$ in this specification is expanded to the machine type. This is unlike other properties, like MDBODY, where a single \$ is expanded to the machine type. If a backslash (\) or a newline must be included literally in the specification, it must be quoted with backslash. |
| <dep1> | |
| <dep2> | |
| ... | |
| <depN> | Are source files, that the <objectfile> depends on. |
| <command> | is a command (sequence) that is executed by the compiler as it is, except for the following substitutions: \$\$ is expanded to the machine type, as explained above. \$0 is expanded to <objectfile> \$1 is expanded to <dep1> |

\$2 is expanded to <dep2>

...

\$N is expanded to <depN>

If a backslash (\) or a newline must be included literally in the commands, it must be quoted with backslash.

Functionality

The <objectfile> is included in the link directive. The compiler will execute <commands> if and only if

- a. <objectfile> does not exist

or

- b. any of the files <dep1>, <dep2>, ... <depN> are newer than an existing <objectfile>

The compiler will execute <commands> from the directory in which the file containing the BUILD property resides.

Example

If the object file foo.o (foo.obj) is to be generated from the foo.c file in the external directory, but also depends on the foo.h file in the external directory, you could specify this as:

```
BUILD nti      '$$/foo.obj' 'external/foo.c' 'external/foo.h'
              'cl -c $1 -Fo$0 -nologo -w -O2 -Zd -Zp4'
ppcmac       '::$$:foo.obj' ':external:foo.c' ':external:foo.h'
              'MrC -D MAC -o $0 $1'
default      '$$/foo.o' 'external/foo.c' 'external/foo.h'
              '$CC -o $1 $2';
```

Notice, that regular environment variables may be used in the <commands> specification, e.g. in the default (UNIX) specification, the variable CC are used (on UNIX, this is always set to an appropriate value in the job-file).

More than one build pr. fragment

In general, more than one build pr. fragment will not work. The reason is that the meta-programming system combines all build directives into one directive (property).

This means that:

```
BUILD sun4s 'cc1' default 'cc2';
BUILD sun4s 'cc3' default 'cc4'
```

means the same as:

```
BUILD sun4s 'cc1' default 'cc2' sun4s 'cc3' default 'cc4'
```

BUILD 'executes' all entries for a given platform. This means that the 2 sun4s entries will be executed for sun4s. The 2 default entries will be executed for all other platforms.

If the following two build entries are used:

```
BUILD sun4s 'cc1' default 'cc2';
BUILD sgi   'cc3' default 'cc4'
```

ONLY cc1 will be executed for sun4s and ONLY cc3 for sgi, and the 2 default entries for all other platforms. This is probably not what is intended: For sun4s you would expect cc1 and cc4 to be executed and sgi, cc2 and cc3.

10 Code Generation for Multiple Machines

When instantiating a configuration for some machine, a number of object files are produced by the compiler - one for each fragment contributing to the configuration. On most architectures, the compiler actually generates symbolic assembly code, and this code is turned into object files by means of the native assembler. The native linker is used to produce an executable image for the machine in question on basis of these object files.

10.1 Placement of Object Code

Different machines normally use different formats for object files. The files containing object code and symbolic assembly code are always placed in a sub-directory relative to the directory containing the common source code. A sub-directory is created for each special object file format. Currently the following subdirectories are used:

| | |
|----------------------|--------------------------------------|
| <code>sun4s</code> | SUN-4 (SPARC) running Solaris |
| <code>hpux9pa</code> | HP 9000/700 running HP-UX |
| <code>sgi</code> | Silicon Graphics (MIPS) running IRIX |
| <code>linux</code> | PC running Linux |
| <code>nti</code> | PC running Windows NT or Windows 95 |
| <code>ppcmac</code> | PowerMacintosh |

For executable images to be activated “directly”, without prefixing their name with the name of a sub-directory, executable images are placed in the same directory as the common source files. It is however possible to control the naming of the executable images. This is done by means of the `-o` option to the compiler.

10.2 Macro Expansion

Consider this use of the `MDBODY` property:

```
MDBODY default './$/betaenvbody_$$'
```

The symbol `$$` is expanded by the compiler. It is expanded to the name of the subdirectory into which the generated code will be placed. That is, if code is generated for a `ppcmac` (Macintosh) machine, the above expands to `./ppcmac/betaenvbody_mac`.

This may be a convenient short-hand, but may also make is possible to instantiate configurations for new machines without changing the original source code.

Bibliography

- [KLMM 94] J. L. Knudsen, M. Löfgren, O. L. Madsen, B. Magnusson (eds.): *Object-Oriented Environments – The Mjølner Approach*, Prentice Hall, 1994, ISBN 0-13-009291-6.
- [MMN93] O. L. Madsen, B. Møller-Pedersen, K. Nygaard: *Object-Oriented Programming in the BETA Programming Language*, Addison-Wesley, 1993, ISBN 0-201-62430-3
- [MIA 90-4] Mjølner Informatics: *The Mjølner System: Using BETA on UNIX Systems*, Mjølner Informatics Report MIA 90-4.
- [MIA 90-6] Mjølner Informatics: *The Mjølner System: Using BETA on the Macintosh*, Mjølner Informatics Report MIA 90-6.
- [MIA 90-8] Mjølner Informatics: *The Mjølner System: Basic Libraries, Reference Manual*, Mjølner Informatics Report MIA 90-8
- [MIA 90-11] Mjølner Informatics: *Sif – A Hyper Structure Editor, Tutorial and Reference Manual* Mjølner Informatics Report MIA 90-11.
- [MIA 92-12] Mjølner Informatics: *The Mjølner System – The BETA Source-level Debugger – Users's Guide*, Mjølner Informatics Report MIA 92-12
- [MIA 94-24] Mjølner Informatics: *The Mjølner System – The Mjølner System Tutorial* Mjølner port MIA 94-24.
- [MIA 94-34] Mjølner Informatics: *The Mjølner System – Using on Windows 95 or Windows NT* Mjølner port MIA 94-34.

Appendix A. Semantic Errors and Warnings

A.1 Semantic Errors

The following is a list of semantic error messages that may be reported by the compiler. See also section 7.2.

1. Name is declared more than once
2. Name is not declared
3. Attribute is not declared
4. A pattern is expected here
5. An item is expected here
6. A repetition is expected here
7. A simple evaluation cannot be assigned
8. The lists have different lengths
9. The lists have different lengths
10. In "leave P" or "restart P", "P" must be an enclosing label or enclosing pattern
11. Illegal assignment/comparison of value, reference or repetition
12. Only a single name is allowed here
13. Attempt to bind V which is not virtual (V ::< T)
14. In V ::< T, T does not have a correct qualification
15. An object is expected here
16. A basic pattern cannot be used as a super-pattern
17. A virtual pattern or a pattern defined as a descriptor slot cannot be used as super-pattern
18. A string of length 1 is a char - NOT a text
19. Illegal recursion in the definition of a pattern.
One of the following type of errors have occurred:
 - (1) There may be a circle in the super-pattern chain:
A: C(# ... #); B: A(# ... #); C: B(# ... #)
 - (2) The pattern may direct or indirectly contain a static instance of itself:

P: (# ...; X: @P; ... do ... #)

(3) The pattern may directly or indirectly contain an inserted instance of itself:

P: (# ... do ...; P(# ... #); ... #) or

A: (# ... P: (# R: ^A; ... do ...; R.P(# ... #); ... #) ... #)

20. Incompatible qualifications in assignment/comparison
21. Only simple values or references may be compared
22. Only simple values may appear in unary expressions
23. Fatal error: virtual binding not found
27. The descriptor is both used as item and component
28. Static size of descriptor is larger than 32760 bytes
29. Illegal recursion in object-description
30. Illegal assignment to constant value/reference or repetition
31. Only pattern-declarations may appear in a fragment of category 'attributes'
32. A virtual qualification must be a pattern name or a descriptor
33. A virtual pattern or descriptor-slot cannot be used as a component
34. An enter/exit parameter of an "external" must be one of:
integer,char,real,integer-repetitions,char-repetition,
subpattern of cstruct,variable-subpattern of external
35. An "external" can only have one exit parameter
36. A sub-pattern of "external" cannot be used as super-pattern
37. The DO-part of an "external" should be empty
38. A repetition/for-imp range must be an integer, char or boolean evaluation
39. A simple pattern cannot be used here
40. Unknown inline primitive
41. The superpattern of this descriptor has no INNER
42. Attempt to bind a virtual in a descriptor with no superpattern
43. The qualification of a variable pattern must be a pattern
44. A pattern-, virtual-, variable-pattern, or reference is expected here
45. A repetition name is expected here
46. In "this(P)" or "inner P", P must be the name of an enclosing pattern
47. An unexpanded nonterminal must be a SLOT
48. A super-pattern must be a simple pattern or a simple
pattern attribute of a static object
49. A simple pattern or virtual pattern cannot be assigned
a structure reference
50. A structure reference can only be assigned to/compared
with another structure reference
51. Only integer,char,boolean, real objects and references can be
compared in an if-imperative
52. Rename declaration has NOT been implemented
53. Syntax error in number
54. Name not declared. There is no corresponding virtual declaration
55. A pattern with a do-part slot cannot be used as a super-pattern

56. The QUA construct has not been implemented
57. A basic pattern like integer, real, char, boolean, false, and true cannot be used as a super-pattern
58. In a list being assigned to and being assigned from as in
...->(E1,E2,...En)-> ...
the elements may not be patterns
59. The enter-parameters of an external call must be supplied
60. The left-side of the assignment/comparison has no (exit-)list or the right-side has no (enter-)list
61. An element of the left-side/right-side of the assignment/comparison has no (exit-)list or (enter-)list
62. The Left-side of the assignment/comparison has no (exit-)list
63. An element of the left-side of the assignment/comparison has no (exit-)list
64. The right-side of the assignment/comparison has no (enter-)list
65. An element of the right-side of the assignment/comparison has no (enter-)list
66. A simple value (integer,boolean,char,real) cannot be assigned/compared to/with a list
67. An object with no exit-list is being assigned/compared to a reference.
The left-side may be missing a "[]" or the right-side may have a superfluous "[]"
68. An element with no exit-list in the left-side list is being assigned/compared to a reference on the right-side
The left-side may be missing a "[]" or the right-side may have a superfluous "[]"
69. A reference is being assigned/compared to an object with no enter-list
The right-side may be missing a "[]" or the left-side may have a superfluous "[]"
70. A reference is being assigned/compared to an element on the right-side with no enter-list
The right-side may be missing a "[]" or the left-side may have a superfluous "[]"
80. "inner P" is only legal in the do-part of the pattern "P"
81. In a computed-remote, "(EV).X", "EV" cannot be an evaluation-list
82. In a computed-remote, "(EV).X", "EV" must have one exit-element, which must be a reference
83. In a computed-remote, "(EV).X", "EV" is not a legal evaluation
84. "Extend" and "new" must have an enter-parameter
85. "leave P" or "restart P", where "P" is a pattern, is only legal in the do-part of "P"
87. A repetition index must be an integer-evaluation
88. The base of this number is too large
89. A subpattern of "data" may only have declarations of the forms:
"X: ^T" where "T" is subpattern of "data", or
"X: @T" where "T" is integer,shortint,char,boolean,real
or subpattern of "data"
90. A subpattern of "data" may not have a do-part
91. A boolean evaluation is expected here
92. Primitive operation appears in wrong context
93. It is not possible to obtain a structure reference for a basic pattern like integer, real, char, boolean, false, and true or instances of these
94. A virtual pattern cannot be bound to a basic pattern like

- integer, real, char, boolean, false, and true
96. In "X: ^<AD>.P", "Y: @<AD>.P", "<AD>" cannot be:
 - a repetition element as in "R[e].P"
 - a computed remote as in "(R).P"
 It must be a static object
 98. A sub-pattern of "external" must be defined as a pattern
 100. In "V :: T", "T" must be a non-virtual pattern
 101. In "V :: T", "T" must be a pattern
 102. A cycle has been detected in the super-chain of the virtual/final binding
 103. Incompatible types of binary operator
 104. Incompatible left- and right-side of assignment
 105. Illegal assignment to constant, literal or expression
 107. A virtual cannot be bound to a slot
 108. Illegal use of the "&"-operator
 110. Illegal recursion in exit list:
 - a pattern is referred directly or indirectly in its own exit list
 111. Illegal recursion in enter list:
 - a pattern is referred directly or indirectly in its own enter list
 112. External entry point has a blank- or control character
 113. There is a circle in the super-pattern chain
 114. Illegal operator "!"

A.2 Semantic Warnings

24. A run-time qualification check will be generated here
25. Repetition of static components is not implemented
26. Repetition of non simple patterns is not implemented
86. "leave P" and "restart P", where "P" is a pattern,
 - are currently not allowed in internal descriptors of "P"
95. In "X: ^R.P", "Y: @R.P", or "Z: @R.P(#...#)",
 - "R" should NOT be a dynamic reference!
 - For "X: ^R.P", consider using "X: ^T.P",
 - where "T" is the pattern qualifying "R" ("R: ^T").
 - A future release may consider this to be a semantic error.
97. An "inner" in a singular object will never be executed
99. Final binding to a virtual pattern is a new facility
 - in this version of the compiler.
 - Please report any problems to support@mjolner.dk
106. Assignment/comparison between boolean and integer
109. Text has a null-char. All chars after the null-char are ignored
115. Use of "@@" in combination with object executions in external calls is insecure'

Appendix B. The BETA Grammar

This appendix contains a listing of a grammar describing the BETA language accepted by the compiler. The grammar formalism used in the Mjølnir System is a variant of context free grammars. A *structured context free grammar* is a context free grammar (CFG) where the rules (productions) satisfy a certain structure. See [MIA90-8] for a description of structured context free grammars.

```

<BetaForm>          ::= | <DescriptorForm>
                   | <AttributesForm>
<DescriptorForm>   ::= <ObjectDescriptor>
<AttributesForm>   ::= <Attributes>
<ObjectDescriptor> ::= <PrefixOpt> <MainPart>
<MainPart>         ::= '(' <Attributes> <ActionPart> '#'
<Attributes>       ::=+ <AttributeDeclOpt> ';'
<PrefixOpt>        ::= ? <Prefix>
<Prefix>           ::= <AttributeDenotation>
<AttributeDeclOpt> ::= ? <AttributeDecl>
<AttributeDecl>    ::= | <PatternDecl>
                   | <SimpleDecl>
                   | <RepetitionDecl>
                   | <VirtualDecl>
                   | <BindingDecl>
                   | <FinalDecl>
<PatternDecl>      ::= <Names> ':' <ObjectDescriptor>
<SimpleDecl>       ::= <Names> ':' <referenceSpecification>
<RepetitionDecl>  ::= <Names> ':' '[' <index> ']' <referenceSpecification>
<VirtualDecl>     ::= <Names> ':' '<' <ObjectSpecification>
<BindingDecl>     ::= <Names> ':' ':' '<' <ObjectSpecification>
<FinalDecl>       ::= <Names> ':' ':' <ObjectSpecification>
<VariablePattern> ::= '##' <AttributeDenotation>
<referenceSpecification> ::= | <StaticItem>
                           | <DynamicItem>
                           | <StaticComponent>
                           | <DynamicComponent>
                           | <VariablePattern>
<StaticItem>       ::= '@' <ObjectSpecification>
<DynamicItem>     ::= '^' <AttributeDenotation>
<StaticComponent> ::= '@' '|' <ObjectSpecification>
<DynamicComponent> ::= '^' '|' <AttributeDenotation>
<ObjectSpecification> ::= | <ObjectDescriptor>
                       | <AttributeDenotation>
<Index>           ::= | <SimpleIndex>
                   | <NamedIndex>
<NamedIndex>      ::= <NameDcl> ':' <Evaluation>
<ActionPart>     ::= <EnterPartOpt> <DoPartOpt> <ExitPartOpt>
<EnterPartOpt>   ::= ? <EnterPart>
<DoPartOpt>      ::= ? <DoPart>
<ExitPartOpt>    ::= ? <ExitPart>
<EnterPart>      ::= 'enter' <Evaluation>
<DoPart>         ::= 'do' <Imperatives>

```

```

<ExitPart> ::= 'exit' <Evaluation>
<Imperatives> ::=+ <ImpOpt> ';'
<ImpOpt> ::=? <Imp>
<Imp> ::= |
        | <LabelledImp>
        | <ForImp>
        | <SimpleIfImp>
        | <GeneralIfImp>
        | <LeaveImp>
        | <RestartImp>
        | <InnerImp>
        | <SuspendImp>
        | <Evaluation>
<LabelledImp> ::= <NameDcl> ':' <Imp>
<ForImp> ::=
        '(' 'for' <Index> 'repeat' <Imperatives> 'for' ')'
<GeneralIfImp> ::=
        '(' 'if' <Evaluation> <Alternatives> <ElsePartOpt> 'if' ')'
<SimpleIfImp> ::=
        '(' 'if' <Evaluation> 'then' <Imperatives> <ElsePartOpt> 'if' ')'
<LeaveImp> ::= 'leave' <NameApl>
<RestartImp> ::= 'restart' <NameApl>
<InnerImp> ::= 'inner' <NameAplOpt>
<NameAplOpt> ::=? <NameApl>
<SuspendImp> ::= 'suspend'
<Alternatives> ::=+ <Alternative>
<Alternative> ::= <Selections> 'then' <Imperatives>
<Selections> ::=+ <Selection>
<Selection> ::= |
        | <CaseSelection>
<CaseSelection> ::= '/' '/' <evaluation>
<ElsePartOpt> ::=? <ElsePart>
<ElsePart> ::= 'else' <Imperatives>
<Evaluations> ::=+ <Evaluation> ','
<Evaluation> ::= |
        | <Expression>
        | <AssignmentEvaluation>
<AssignmentEvaluation> ::= <Evaluation> '->' <Transaction>
<Transaction> ::= |
        | <ObjectEvaluation>
        | <ObjectReference>
        | <EvalList>
        | <StructureReference>
<ObjectEvaluation> ::= |
        | <InsertedItem>
        | <reference>
<Reference> ::= |
        | <ObjectDenotation>
        | <DynamicObjectGeneration>
<DynamicObjectGeneration> ::= |
        | <DynamicItemGeneration>
        | <DynamicComponentGeneration>
<InsertedItem> ::= <ObjectDescriptor>
<ObjectDenotation> ::= <AttributeDenotation>
<ObjectReference> ::= <Reference> '['
<StructureReference> ::= <AttributeDenotation> '##'
<EvalList> ::= '(' <Evaluations> ')'
<DynamicItemGeneration> ::= '&' <ObjectSpecification>
<DynamicComponentGeneration> ::= '&' '|' <ObjectSpecification>

<AttributeDenotation> ::= |
        | <NameApl>
        | <Remote>
        | <ComputedRemote>
        | <Indexed>
        | <ThisObject>
<Remote> ::= <AttributeDenotation> '.' <NameApl>
<ComputedRemote> ::= '(' <Evaluations> ')' '.' <NameApl>
<Indexed> ::= <AttributeDenotation> '[' <Evaluation> ']'
<ThisObject> ::= 'this' '(' <NameApl> ')'
<Expression> ::= |
        | <RelationalExp> | <SimpleExp>
<RelationalExp> ::= |
        | <EqExp> | <LtExp> | <LeExp>
        | <GtExp> | <GeExp> | <NeExp>
<SimpleExp> ::= |
        | <AddExp> | <SignedTerm> | <Term>
<AddExp> ::= |
        | <PlusExp> | <MinusExp> | <OrExp> | <XorExp>

```

```

<SignedTerm>      ::= | <unaryPlusExp> | <unaryMinusexp>
<Term>           ::= | <MulExp> | <Factor>
<MulExp>         ::= | <TimesExp> | <RealDivExp> | <IntDivExp>
                  | <ModExp> | <AndExp>
<EqExp>          ::= <Operand1:SimpleExp> '=' <Operand2:SimpleExp>
<LtExp>          ::= <Operand1:SimpleExp> '<' <Operand2:SimpleExp>
<LeExp>          ::= <Operand1:SimpleExp> '<=' <Operand2:SimpleExp>
<GtExp>          ::= <Operand1:SimpleExp> '>' <Operand2:SimpleExp>
<GeExp>          ::= <Operand1:SimpleExp> '>=' <Operand2:SimpleExp>
<NeExp>          ::= <Operand1:SimpleExp> '<>' <Operand2:SimpleExp>
<PlusExp>        ::= <SimpleExp> '+' <Term>
<MinusExp>       ::= <SimpleExp> '-' <Term>
<OrExp>          ::= <SimpleExp> 'or' <Term>
<XorExp>         ::= <SimpleExp> 'xor' <Term>
<unaryPlusExp>   ::= '+' <Term>
<unaryMinusexp> ::= '-' <Term>
<TimesExp>       ::= <Term> '*' <Factor>
<RealDivExp>     ::= <Term> '/' <Factor>
<IntDivExp>      ::= <Term> 'div' <Factor>
<ModExp>         ::= <Term> 'mod' <Factor>
<AndExp>         ::= <Term> 'and' <Factor>
<Factor>         ::= | <TextConst>
                  | <IntegerConst>
                  | <NotExp>
                  | <NoneExp>
                  | <RepetitionSlice>
                  | <Transaction>
<RepetitionSlice> ::= <AttributeDenotation>
                  '[' <Low:Evaluation> ':' <High:Evaluation> ']'
<notExp>         ::= 'not' <factor>
<noneExp>        ::= 'none'
<Names>          ::=+ <NameDcl> ','
<NameDcl>        ::= <NameDecl>
<NameApl>        ::= <NameApl>
<SimpleEntry>    ::= ? <TextConst>
<TextConst>     ::= <String>
<IntegerConst>  ::= <Const>
<SimpleIndex>   ::= <Evaluation>

```


Appendix C. New Features in v5.3

The following new features have been implemented in version 5.3 of the compiler, compared to version 5.1.

C.1 New Platforms

The following new platforms are available:

- The compilers for Windows NT and Windows 95 now generate binary code directly.
- A version for the PowerPC based Macintosh generating binary code is now available.

C2. The BUILD Property

See section 9.1

Index

-, 6

*

*, 6

+

+, 6

.

..db, 5
..job, 5
..s, 5
.ast, 5
.astL, 5
.lst, 5
.o, 5

<

<>, 6

=

=, 6

A

abstract syntax tree, 5
activation stack, 27
and .i.real, 6
ASCII, 6
assembling, 25
assembly code, 5; 34
assembly language, 3
attributes, 13
attributes, 20

B

backslash, 12; 45
backspace, 12; 45
basiclib, 3
BETA language, 6
BETALIB, 4
BETALIB, 17
BETARUN, 22; 31
BETARUN, 18
big-endian, 5
BODY, 22
BODY, 16; 17; 32

C

C, 3
C compiler, 31
call stack, 26
carriage return, 12; 45
, 45
command line arguments, 28
Compatibility, 7; 8
compiler info output, 29
compiler switches, 29
concurrency, 10
Configurations, 31
context free grammars, 41
context free syntax, 24
cross-compilation, 32

D

data structures, 19
debug info, 29
, 29
descriptor, 13; 20; 25; 45
Deviations, 6
deviations, 1
disk space, 25
div, 6
DoPart, 13
DoPart, 25; 45
DoPart slot, 20
double quote, 12; 45
dynamic item, 9
dynamic linking, 28

E

extension, 17
 External library, 18

F

false, 6
 file name, 17
 File Name Restrictions, 23
 Files Generated by the Compiler, 5
final binding, 9
 formal syntax of the BETA fragment-system, 22
 formfeed, 12; 45
 fragment, 13
 Fragment Denotations, 17
 fragment system, 2
 fragments, 30

G

grammar, 41

H

horizontal tab, 12; 45

I

If, 12
 implementation, 19
 INCLUDE, 16; 22
INCLUDE, 17
 incomplete file, 25
 INNER, 20
 INNER A, 21
 Inserted components., 9
 inserted item, 10
 Inserted items, 9
 integer, 6
 interface, 13; 19
 , 45

L

label, 10
 labelled compound imperative, 9
 leave, 10
LIBFILE, 18; 22; 31
 library, 3
 linker-directive, 18
 linking, 25
LINKOPT, 18; 22; 31
 LINKOPT, 32
 little-endian, 5
 lst-file, 25

M

MachineSpecificationList, 17; 18; 22; 31
 Macro Expansion, 34
 MAKE, 22; 31
MAKE, 18; 31
 MDBODY, 22; 31
MDBODY, 17; 31; 32
 mod, 6
 modularization, 13
 Modularization of Data Structures, 19
 Multiply Defined Symbol, 25; 45

N

newline, 12; 45
 NONE references, 29

O

object code, 5
 object file, 18
 object files, 34
 objectDescriptor, 13
 OBJFILE, 22; 31
OBJFILE, 18
 octal number, 12; 45
 OFF, 22
OFF, 18
 ON, 22
ON, 18
 ORIGIN, 16; 22
ORIGIN, 17

P

PowerPC, 45
 Private, 16
properties, 16

Q

QUA checks, 29
 question mark, 12; 45

R

real, 6
 recursion, 9
 repeating mode, 4; 28
 RESOURCE, 22; 32
RESOURCE, 18
 restart, 10
restrictions, 1; 23
 result type, 6; 7; 8
 runtime checks for QUA errors, 29
 Run-time errors, 26
 run-time system, 18

S

semantic error messages, 25; 37
Semantic Errors, 37
separate compilation, 13
Short-circuit Boolean Expressions, 12
, 45
Simple If, 12
single quote, 12; 45
SLOT, 25; 45
slot, 2
special characters, 11; 45
static linking, 28
Static Semantic Errors, 24
statically enclosing object, 27
string literal, 11
string literals, 11; 45
struc, 12; 45
structured context free grammar, 41
Syntax Errors, 24
system errors, 26

T

this, 10
trace, 25; 29
true, 6

U

Undefined Reference, 25; 45

V

Valhalla, 5; 29
variant control, 13
vertical tab, 12; 45
Virtual superpatterns, 9

W

warnings, 29

X

xor, 12

~

~beta, 4